

14. РАБОТА С ВРЕМЕННЫМ ХРАНИЛИЩЕМ

С помощью временного хранилища можно хранить картинки и прочие двоичные файлы в информационной базе 1С.

1. Принципы работы с временным хранилищем

Для того чтобы поместить какой-нибудь файл в базу посредством использования временного хранилища, нужно выполнить два шага (см. рис. 14.1 и 14.2):

- На первом шаге загружается файл с файловой системы пользователя в клиентское приложение («тонкий» или web-клиент). После этого файл передается во временное хранилище на кластере. Временное хранилище – это, по сути, служебные файлы, которые хранятся в файловой системе кластера.
- Вторым шагом файл помещается из временного хранилища в серверную среду, а затем в базу данных, где будет храниться в реквизите, имеющем тип *ХранилищеЗначения*.




Рис. 14.1. Помещение файла во временное хранилище



Рис. 14.2 Помещение файла из временного хранилища в БД

Обратная задача решается проще:

- Если нужно взять файл из временного хранилища, то он сначала перемещается в серверную среду (кластер серверов 1С), потом в клиентское приложение («тонкий» или web-клиент), а уже потом в файловую систему (см. рис. 14.3).
- Если же нужно извлечь файл из базы данных, то он, минуя временное хранилище, уходит в серверную среду, потом в клиентское приложение, а потом в файловую систему (см. рис. 14.4).




Рис. 14.3. Извлечение файла из временного хранилища




Рис. 14.4. Извлечение файла из базы данных

Все примеры рассматриваются в режиме отключенной модальности, т.е. с использованием асинхронных вызовов.

Пример 1.

Организовать хранение различных файлов форматов docx и xlsx контрагентов в специальном справочнике.

При организации структуры хранения файлов всегда лучше использовать отдельный справочник или регистр сведений.

- 1) Создать подчиненный справочнику **Контрагенты** справочник **ФайлыКонтрагента** с реквизитом **Файл** типа **ХранилищеЗначения**.




Рис. 14.5. Реквизит «Файл» справочника «ФайлыКонтрагента»

- 2) Включить новый справочник в подсистемы **Закупки** и **Продажи**.
- 3) Для ролей **Администратор**, **Работа с закупками** и **Работа с продажами** установить полные права на этот справочник (кроме интерактивного удаления), для остальных ролей установите только право на чтение.
- 4) Добавить форму элемента справочника **ФайлыКонтрагента**.
- 5) На форме создать две команды:
 - **ЗагрузитьФайл** – будет загружать файл с жесткого диска,
 - **СохранитьФайл** – будет сохранять файл на жесткий диск (см. рис. 14.6).




Рис. 14.6. Команды для загрузки и сохранения файлов

- 6) Добавить на форму реквизит **СсылкаВоВременномХранилище**, но не размещать его в виде поля на самой форме (см. рис. 14.7).




Рис. 14.7. Реквизит **СсылкаВоВременномХранилище**

2. Загрузка файлов в базу данных

Для обработки загрузки файла используется метод **НачатьПомещениеФайла** (помещает файл во временное хранилище).

Синтаксис метода:

НачатьПомещениеФайла(<ОписаниеОповещенияОЗавершении>, <Адрес>, <НачальноеИмяФайла>, <Интерактивно>, <УникальныйИдентификаторФормы>)

Где:

ОписаниеОповещенияОЗавершении – описание оповещения, указывающая экспортную процедуру, которая выполнится после помещения файла. Эта процедура должна иметь следующие параметры:

- Результат – «Истина» или «Ложь», если параметр Интерактивно – «Истина». *Ложь* – значит пользователь отказался выбрать файл;
- Адрес – навигационная ссылка расположения нового файла;
- ВыбранныйФайл – путь к файлу, указанный в диалоге выбора;
- ДополнительныеПараметры – если были указаны при создании оповещения.

Адрес – адрес файла во временном хранилище (навигационная ссылка), в который будет помещен файл, если не указан или пустой, то будет создан новый.

НачальноеИмяФайла – в том случае, если *Интерактивно* – *Ложь*, нужно указать полный путь к файлу, который нужно поместить во временное хранилище.

Интерактивно – если *Истина*, то будет вызван диалог выбора файла.

УникальныйИдентификаторФормы – уникальный идентификатор текущей формы.

- 7) Написать код в обработчике команды **ЗагрузитьФайл** (см. листинг 14.1).

&НаКлиенте

Процедура ЗагрузитьФайл(Команда)

ОписаниеОповещения = Новый ОписаниеОповещения("ПослеПомещенияФайла", ЭтаФорма);

НачатьПомещениеФайла(ОписаниеОповещения,,, Истина, УникальныйИдентификатор);

КонецПроцедуры

&НаКлиенте

Процедура ПослеПомещенияФайла(Результат, Адрес, ВыбранноеИмяФайла, ДопПараметры) Экспорт

Если Не Результат Тогда

Возврат;

КонецЕсли;

СсылкаВоВременномХранилище = Адрес;

Файл = Новый Файл(ВыбранноеИмяФайла);

Объект.Наименование = Файл.Имя;

Модифицированность = Истина;

КонецПроцедуры

Листинг 14.1. Помещение файла во временное хранилище

- 8) Запустить процедуру **ЗагрузитьФайл**, откроется диалог выбора файла (см. рис. 14.8).




Рис. 14.8. Диалог выбора файла

После выбора файла (или отказа от выбора) сработает процедура **ПослеПомещенияФайла**, в которой выполняется проверка того, выбран файл или нет.

Один из параметров процедуры **Адрес** – содержит адрес временного хранилища, в котором содержится выбранный файл. Этот адрес сохраняется в реквизит формы **СсылкаВоВременномХранилище** (он нужен, чтобы сохранить файл из временного хранилища в базу данных).

Также в наименование элемента справочника записывается короткое имя файла (см. рис. 14.9), используя объект **Файл**.




Рис. 14.9. После выбора файла

В диалоге выбора файла можно выбрать любой файл. Но если требуется наложить фильтр на выбор файла, то в процедуре **ЗагрузитьФайл** нужно использовать объект **ДиалогВыбораФайла**.

- 9) Изменить код обработчика для установки фильтра на выбор нужного расширения (см. рис. 14.10).

&НаКлиенте

Процедура ЗагрузитьФайл (Команда)

```
Режим = РежимДиалогаВыбораФайла.Открытие;
ДиалогОткрытия = Новый ДиалогВыбораФайла (Режим);
ДиалогОткрытия.ПолноеИмяФайла = "";
Фильтр = "Файл Word (*.docx) | *.docx | Файл Excel (*.xlsx) | *.xlsx";
ДиалогОткрытия.Фильтр = Фильтр;
ДиалогОткрытия.МножественныйВыбор = Ложь;
ДиалогОткрытия.Заголовок = "Выберете файл для загрузки";
```

```
ОписаниеОповещения = Новый ОписаниеОповещения ("ПослеЗагрузкиФайла", ЭтаФорма);
```

```
ДиалогОткрытия.Показать (ОписаниеОповещения);
```

КонецПроцедуры

&НаКлиенте

Процедура ПослеЗагрузкиФайла (ВыбранныйФайл, ДопПараметр) Экспорт

```
Если ВыбранныйФайл = Неопределено Тогда
```

```
    Возврат;
```

```
КонецЕсли;
```

```
ОписаниеОповещения = Новый ОписаниеОповещения ("ПослеПомещенияФайла", ЭтаФорма);
```

```
НачатьПомещениеФайла (ОписаниеОповещения, , ВыбранныйФайл[0], Ложь, УникальныйИдентификатор);
```

КонецПроцедуры

&НаКлиенте

Процедура ПослеПомещенияФайла (Результат, Адрес, ВыбранноеИмяФайла, ДопПараметры) Экспорт

```
Если Не Результат Тогда
```

```
    Возврат;
```

```
КонецЕсли;
```

```
СсылкаВоВременномХранилище = Адрес;
```

```
Файл = Новый Файл (ВыбранноеИмяФайла);
```

```
Объект.Наименование = Файл.Имя;
```

```
Модифицированность = Истина;
```

КонецПроцедуры

Листинг 14.2. Использование диалога выбора файла при помещении файла в хранилище

В этом коде в процедуре *ЗагрузитьФайл* используется объект *ДиалогВыбораФайла*, а само помещение файла в хранилище выполняется уже в процедуре *ПослеЗагрузкиФайла*, которая запустится после того, как пользователь выберет файл в диалоге.

В процедуре используется метод *НачатьПомещениеФайла*, только в этот раз заполняется параметр *НачальноеИмяФайла*, а в параметр *Интерактивно* записывается значение *Ложь*.





Рис. 14.10. Выбор файла с фильтром

После записи файла во временное хранилище его нужно сохранить в саму базу данных. Сделать это можно в обработчике события формы *ПередЗаписьюНаСервере*.

- 10) Создать обработчик события формы *ПередЗаписьюНаСервере* (открыть список выбора *Процедуры и функции* и выбрать нужную процедуру)



- 11) Записать следующий код в обработчике (см. листинг 14.3.

```

&НаСервере
□ Процедура ПередЗаписьюНаСервере (Отказ, ТекущийОбъект, ПараметрыЗаписи)

    Если ЭтоАдресВременногоХранилища (СсылкаВоВременномХранилище) Тогда
        Данные = ПолучитьИзВременногоХранилища (СсылкаВоВременномХранилище) ;
        ТекущийОбъект.файл = Новый ХранилищеЗначения (Данные) ;
        УдалитьИзВременногоХранилища (СсылкаВоВременномХранилище) ;
    КонецЕсли;

КонецПроцедуры

```

Листинг 14.3. Загрузка из временного хранилища

Ход работы процедуры-обработчика:

1. Проверяется, является ли реквизит формы *СсылкаВоВременномХранилище* адресом временного хранилища. Делается это при помощи метода глобального контекста *ЭтоАдресВременногоХранилища*, который возвращает «Истина», если строка, которая в его параметре, является временным хранилищем.
 2. Если адрес есть, нужно получить данные из временного хранилища при помощи функции *ПолучитьИзВременногоХранилища*, в качестве параметра которой указывается адрес файла во временном хранилище.
 3. И поскольку реквизит *Файл* справочника имеет тип *ХранилищеЗначения*, то значение этого типа и нужно записать в этот реквизит. Для этого создается объект *ХранилищеЗначения* при помощи конструктора *Новый*, где в качестве параметра указывается файл, который будет в хранилище.
 4. И в конце удаляется файл из временного хранилища.
- В результате файл буде записан в базу данных.

Получение данных из временного хранилища нужно выполнять в серверном контексте!

3. Сохранение файла из базы данных

Для сохранения файла из базы данных на жесткий диск компьютера нужно использовать метод *НачатьПолучениеФайлов*, который имеет следующий синтаксис:

**НачатьПолучениеФайлов(<ОписаниеОповещения>,<ПолучаемыеФайлы>,
<РасположениеФайлов>,<Интерактивно>)**

Где:

ОписаниеОповещения – описание оповещения, которое содержит процедуру с двумя параметрами:

- «ПолученныеФайлы» – массив объектов с типом *ОписаниеПереданногоФайла* (в этом типе хранятся навигационные ссылки полученных файлов, и полный путь, куда они были сохранены), или тип «Неопределенно», если ничего не было получено;
- «ДополнительныеПараметры» – если были указаны при создании оповещения.

ПолучаемыеФайлы – массив объектов *ОписаниеПереданногоФайла*, в каждом элементе которого хранится описание сохраняемого файла.

Объект *ОписаниеПереданногоФайла* имеет два свойства:

- *Хранение* – это или адрес в информационной базе, или адрес во временном хранилище,
- *Имя* – путь, где расположен файл, может быть, имя файла.

РасположениеФайлов – это или диалог выбора файла, или строка, куда файл будет сохранен.

Интерактивно – «Ложь» или «Истина» (файл будет сохраняться с выбором диалога, или нет).

12) В процедуре обработчике команды **Сохранить файл** записать следующий код (см. листинг 14.4).

```

    &НаКлиенте
    □ Процедура СохранитьФайл (Команда)

        СсылкаНаФайл = ПолучитьНавигационнуюСсылку (Объект.Ссылка, "файл") ; //1
        Если ПустаяСтрока (СсылкаНаФайл) Тогда
            Возврат;
        КонецЕсли;
        //2
        ПолучаемыйФайл = Новый ОписаниеПередаваемогоФайла;
        ПолучаемыйФайл.Имя = Объект.Наименование;
        ПолучаемыйФайл.Хранение = СсылкаНаФайл;
        //3
        МассивФайлов = Новый Массив;
        МассивФайлов.Добавить (ПолучаемыйФайл) ;
        //4
        Режим = РежимДиалогаВыбораФайла.Сохранение;
        ДиалогСохранения = Новый ДиалогВыбораФайла (Режим) ;
        //5
        ОписаниеОповещения = Новый ОписаниеОповещения ("ПослеПолученияфайла", ЭтаФорма) ;
        //6
        НачатьПолучениеФайлов (ОписаниеОповещения, МассивФайлов, ДиалогСохранения, Истина) ;

    КонецПроцедуры

    &НаКлиенте
    □ Процедура ПослеПолученияфайла (ПолученныеФайлы, ДопПараметры) Экспорт

    КонецПроцедуры
```

Листинг 14.4. Получение файла

Пояснения к коду:

Поскольку сам файл хранится в реквизите справочника **Файл**, то нужно получить навигационную ссылку на реквизит этого объекта, что делается в строке //1.

В блоке //2 создается объект *ОписаниеПередаваемогоФайла*, в свойство *Имя* этого объекта записывается наименование элемента справочника **ФайлыКонтрагента**, а в свойство *Хранение* записывается навигационная ссылка на файл, который был получен в строке //1.

В блоке //3 созданный в блоке //2 объект *ПолучаемыйФайл* с типом *ОписаниеПередаваемогоФайла* записывается в массив, необходимый для метода *НачалоПолучениеФайлов*.

В блоке //4 создается объект с типом *ДиалогСохраненияФайла*, который используется в качестве параметра *РасположениеФайлов* метода *НачалоПолучениеФайлов*.

В строке //5 создается описание оповещения, его необходимо использовать даже в том случае, если после сохранения файла ничего не нужно делать (при этом экспортная процедура этого оповещения будет пустая).

В строке //6 используется метод *НачалоПолучениеФайлов*, где указывается описание оповещения, массив файлов, в котором есть один элемент с типом «ОписаниеПередаваемогоФайла», содержащий навигационную ссылку на реквизит «Файл справочника».

13) Проверить работу команды **Сохранить файл** (см. рис. 14.11).




Рис. 14.11. Сохранение файла

4. Работа с картинками

Алгоритм работы с картинками аналогичен работе с файлами:

- в специально созданном для хранения картинок справочнике или регистре сведений создается реквизит с типом *ХранилищеЗначения*, где будет храниться файл-картинка,
- на форме создается реквизит с типом *Строка* для хранения навигационной ссылки на картинку (в базе или во временном хранилище);
- при загрузке картинка помещается во временное хранилище при помощи метода *НачатьПолучениеФайло*;
- при сохранении картинки в базе она извлекается из временного хранилища и при помощи объекта *ХранилищеЗначения* записывается в реквизит справочника или регистра;
- при открытии формы навигационная ссылка на картинку извлекается из реквизита объекта и присваивается реквизиту формы.

Пример 2.

Организовать в справочнике **Номенклатура** хранение картинок с изображениями товаров.

- 1) В справочнике **Номенклатура** создать новый реквизит **Картинка** с типом *ХранилищеЗначения* (см. рис. 14.12).




Рис. 14.12. Реквизит «Картинка» справочника «Номенклатура»

- 2) Разместить на форме элемента справочника **Номенклатура** реквизит **СсылкаНаКартинку** с типом *Строка* и поместить этот реквизит на форму в виде поля типа *Поле картинки* (см. рис. 14.13).




Рис. 14.13. Поле формы с видом «Поле картинки»

- 3) Для осуществления загрузки картинки по клику мышки на саму картинку включить в свойствах поля свойство *Гиперссылка* (см. рис. 14.14).




Рис. 14.14. Свойство «Гиперссылка» поля «СсылкаНаКартинку»

- 4) Для поля *СсылкаНаКартинку* создать обработчик события *Нажатия* на клиенте (см. рис. 14.15).




Рис. 14.15. Событие «Нажатие» поля «СсылкаНаКартинку»

- 5) В обработчике этого события написать код, где будет вызываться диалог открытия файла (см. листинг 14.5).

&НаКлиенте

Процедура СсылкаНаКартинкуНажатие(Элемент, СтандартнаяОбработка)

СтандартнаяОбработка = Ложь;

Режим = РежимДиалогаВыбораФайла.Открытие;

ДиалогОткрытия = Новый ДиалогВыбораФайла(Режим);

ДиалогОткрытия.ПолноеИмяФайла = "";

Фильтр = "Файл Jpg (*.jpg)|*.jpg";

ДиалогОткрытия.Фильтр = Фильтр;

ДиалогОткрытия.МножественныйВыбор = Ложь;

ДиалогОткрытия.Заголовок = "Выберете файл для загрузки";

ОписаниеОповещения = Новый ОписаниеОповещения("ПослеЗагрузкиФайла", ЭтаФорма);

ДиалогОткрытия.Показать(ОписаниеОповещения);

КонецПроцедуры

Листинг 14.5. Обработка нажатия на поле

- 6) Описать процедуру *ПослеЗагрузкиФайла*, указанную в описании оповещения. (в этой процедуре в случае выбора пользователем файла он будет помещаться во временное хранилище при помощи метода *НачатьПомещениеФайла*).

&НаКлиенте

Процедура *ПослеЗагрузкиФайла*(*ВыбранныйФайл*, *ДопПараметр*) Экспорт

Если *ВыбранныйФайл* = Неопределено Тогда

Возврат;

КонецЕсли;

ОписаниеОповещения = Новый *ОписаниеОповещения*("ПослеПомещенияФайла", *ЭтаФорма*);

НачатьПомещениеФайла(*ОписаниеОповещения*, *ВыбранныйФайл*[0], *Ложь*, *УникальныйИдентификатор*);

КонецПроцедуры

Листинг 14.6. Процедура «ПослеЗагрузкиФайла»

- 7) В процедуре *ПослеПомещенияФайла*, указанной в описании оповещения процедуры *ПослеЗагрузкиФайла*, присвоить реквизиту формы *СсылкаНаКартинку* адрес временного хранилища (см. листинг 14.7).

&НаКлиенте

Процедура *ПослеПомещенияФайла*(*Результат*, *Адрес*, *ВыбранноеИмяФайла*, *ДопПараметры*) Экспорт

Если Не *Результат* Тогда

Возврат;

КонецЕсли;

СсылкаНаКартинку = *Адрес*;

Модифицированность = *Истина*;

КонецПроцедуры

Листинг 14.7. Процедура «ПослеПомещенияФайла»

- 8) Для сохранения картинки в реквизите объекта создать процедуру обработки события формы *ПередЗаписьюНаСервере*

&НаСервере

Процедура *ПередЗаписьюНаСервере*(*Отказ*, *ТекущийОбъект*, *ПараметрыЗаписи*)

Если ЭтоАдресВременногоХранилища(*СсылкаНаКартинку*) Тогда

ФайлКартинки = *ПолучитьИзВременногоХранилища*(*СсылкаНаКартинку*);

ТекущийОбъект.Картинка = Новый *ХранилищеЗначения*(*ФайлКартинки*);

УдалитьИзВременногоХранилища(*СсылкаНаКартинку*);

СсылкаНаКартинку = *ПолучитьНавигационнуюСсылку*(*Объект.Ссылка*, "Картинка");

КонецЕсли;

КонецПроцедуры

Листинг 14.8. Картинка записывается в реквизит объекта при сохранении

- В начале процедуры проверяется, является ли реквизит формы *СсылкаНаКартинку* адресом временного хранилища, и если это так, то файл картинки получается из временного хранилища и записывается в реквизит *Картинка* объекта.
- Затем картинка удаляется из временного хранилища, а поскольку реквизиту формы был присвоен адрес во временном хранилище, то теперь нужно получить адрес с реквизита объекта, чтобы картинка никуда не делась.

9) В процедуре обработки *При создании формы на сервере* будет получаться навигационная ссылка на картинку (см. листинг 14.9).

&НаСервере

Процедура ПриСозданииНаСервере(Отказ, СтандартнаяОбработка)

СсылкаНаКартинку = ПолучитьНавигационнуюСсылку(Объект.Ссылка,"Картинка");

КонецПроцедуры

Листинг 14.9. Получается навигационная ссылка на картинку

10) В пользовательском режиме после нажатия на картинку откроется форма диалога, где пользователь увидит файлы с расширением jpg (см. рис. 14.16), а при выборе какого-нибудь файла, картинка появится на форме (см. рис. 14.17).




Рис. 14.16. Выбор картинки




Рис. 14.17. Картинка на форме

11) **Самостоятельно** осуществите сохранение картинки при нажатии на саму картинку, при этом должен появиться вопрос *Сохранить картинку?*, и в случае ответа «Да» картинка сохраняется, в случае «Нет» - картинка загружается.