

12. ОТПРАВКА СООБЩЕНИЯ ПОЛЬЗОВАТЕЛЯМ

1. Отправка сообщения пользователю методом Сообщить

Это самый простой метод, который работает в «тонком», «толстом», web-клиенте, на сервере и также в мобильных приложениях. Синтаксис метода:

Сообщить(<ТекстСообщения>, <Статус>)

Где:

ТекстСообщения – текст, который нужно вывести.

Статус – одно из значений системного перечисления «СтатусСообщения», в управляемом приложении этот параметр не работает.

Пример 1.

При проведении документа *Продажа товара* по регистру *Продажи* и по регистру *Остатки товаров по складам* вывести соответствующие сообщения.

- 1) Открыть модуль документа *Продажа товара* и добавить два оператора для вывода сообщений пользователю

```
Процедура ОбработкаПроведения(Отказ, Режим)
    // регистр Продажи
    Движения.Продажи.Записывать = Истина;
    Для Каждого ТекСтрокаТовары Из Товары Цикл
        Движение = Движения.Продажи.Добавить();
    //.....
КонецЦикла;
Сообщить("Провели по РН "Продажи"");

// регистр ОстаткиТоваровНаСкладах Приход
Движения.ОстаткиТоваровНаСкладах.Записывать = Истина;
Для Каждого ТекСтрокаТовары Из Товары Цикл
    Движение = Движения.ОстаткиТоваровНаСкладах.Добавить();
//.....
КонецЦикла;
Сообщить("Провели по РН "Остатки товаров на складах");
КонецПроцедуры
```

Листинг 12.1. Использование сообщений в коде

- 2) В пользовательском режиме создать новый документ *Продажа товара*, провести его и проверить вывод сообщений




Рис. 12.1. Вывод сообщения

2. Вывод окна состояния

Метод окна состояния выводит окно, которое возникает на некоторое время и потом плавно исчезает. Метод работает только в «тонком», «толстом» и web-клиенте.




Рис. 12.2. Окно состояния

Синтаксис метода:

Состояние(<ТекстСообщения>, <Прогресс>, <Пояснение>, <Картинка>)

Где:

ТекстСообщения - строка, которая будет выведена в панель состояния.

Прогресс – число от 1 до 100, значение индикатора прогресса

Пояснение – текст пояснения.

Картинка – картинка, которая будет отображена на окне состояния

Пример 2.

Проиллюстрировать выполнение большого цикла методом окна состояния.

- 1) Добавить новый объект конфигурации – обработку




Рис. 12.3. Добавление обработки

- 2) Добавить форму обработки




Рис. 12.4. Добавление формы обработки

- 3) На форме на вкладке *Команды формы* добавить команду *ПоказатьПрогресс* и переместить ее в элементы формы




Рис. 12.5. Добавление команды на форме обработки

- 4) Создать обработчик команды (в свойствах команды в пункте Действие нажать значок лупы)





Рис. 12.6. Создание обработчика команды

- 5) В созданной процедуре записать код:



- 6) В пользовательском режиме проверить работу команды




Рис. 12.7. Вывод окна состояния с прогрессом

3. Вывод сообщения пользователю

Основной способ вывода сообщения пользователю является отдельным объектом – *СообщениеПользователю*.

Этот объект применим в «тонком», «толстом» и web-клиенте, а также на сервере.

Создается объект при помощи конструктора «Новый».

Пример 3.

Вывести сообщение пользователю при проведении документа *Продажа товара* с помощью объекта *СообщениеПользователю*

- 1) Открыть модуль документа и изменить код

```
// регистр ОстаткиТоваровНаСкладах Приход
Движения.ОстаткиТоваровНаСкладах.Записывать = Истина;
Для Каждого ТекСтрокаТовары Из Товары Цикл
    Движение = Движения.ОстаткиТоваровНаСкладах.Добавить ();
//.....
КонецЦикла;

Сообщение = Новый СообщениеПользователю;
] Сообщение.Текст = "Провели по РН ""Остатки товаров на складах""";
Сообщение.Сообщить ();
```

Листинг 12.3. Используем объект «Сообщение пользователю»

- 2) Проверить, как работает этот код при создании и проведении документа *Продажа товара*.

У этого объекта есть гораздо более интересные функции, например, при проверке заполнения какого-нибудь поля можно выводить сообщение, которое будет указывать на это поле.




Рис. 12.8. Вывод сообщения в поле

Пример 4.

В обработке «Пример2» при выполнении команды, которая запускает прогресс, осуществлять проверку размещённого на форме числа на равенство 0, и если равно, то вывести сообщение, привязанное к этому числу.

- 1) Открыть форму обработки на вкладке *Форма*.
- 2) На вкладке *Реквизиты* в правом окне добавить реквизит с типом *Число* и переместить его на вкладку *Элементы* в правом окне




Рис. 12.9. Реквизит «Число» на форме обработки

- 3) На вкладке *Модуль* дописать в процедуре обработки команды код:

```
НаКлиенте
Процедура ПоказатьПрогресс (Команда)
    Если Число = 0 Тогда
        Сообщение = Новый СообщениеПользователю;
        Сообщение.Текст = "Число не заполнено";
        Сообщение.Поле = "Число";
        Сообщение.Сообщить ();
        Возврат;
    КонецЕсли;
    Для н = 1 по 100000 Цикл
        Если н % 1000 = 0 Тогда
            Состояние ("Н = " + н, н/1000, "Выводим прогресс");
            КонецЕсли;
        КонецЦикла;
    КонецПроцедуры
```

Листинг 12.4. Вывод сообщения, когда число равно 0

- 4) Проверить работу в пользовательском режиме




Рис. 12.10. Вывод сообщения о незаполненном числе

Использовать объект *СообщениеПользователю* можно не только на форме, но и в модуле объекта, например, при проведении документа.

Пример 5.

Реализовать следующую задачу: при проведении документа *Продажа товара* проверять заполнение реквизит *Склад*, и если он не заполнен, то выводить сообщение с привязкой к нужному полю.

- 1) Открыть свойства документа *Продажа товара* и на закладке Данные для реквизита Склад включить проверку заполнения




Рис. 12.11. Проверка заполнения реквизита «Склад»

- 2) Открыть модуль документа и в процедуре проведения перед выполнением всех движений дописать следующий код:

```
Если Не ЗначениеЗаполнено (Склад) Тогда
    Сообщение = Новый СообщениеПользователю;
    Сообщение.Текст = "Не заполнен склад";
    Сообщение.Поле = "Склад";
    Сообщение.ПутьКДанным = "объект";
    Сообщение.КлючДанных = ЭтоОбъект.Ссылка;
    Сообщение.Сообщить ();
Отказ = Истина;
КонецЕсли;
```

Листинг 12.5. Вывод сообщения при проведении документа

- 3) Запустить пользовательский режим и попытаться провести документ *Продажа товара* с пустым полем «Склад»




Рис. 12.12. Ошибка при проведении документа

- В свойство *КлючДанных* объекта *СообщениеПользователю* записывается ссылка на объект, с которым будет связано сообщение.
- Свойство *ПутьКДанным* определяют путь в форме до нужного реквизита. В данном случае это будет основной реквизит формы «Объект».
- Вполне можно было вместо двух этих строк:

```
Сообщение.Поле = "Склад";
Сообщение.ПутьКДанным = "Объект";
```

написать одну строку:

```
Сообщение.Поле = "Объект.Склад";
```

Таким образом, указывая три свойства: «КлючДанных», «ПутьКДанным» и «Поле», определяется:

- ✓ ссылка на объект, где должно выйти сообщение,
- ✓ путь до реквизита формы, на которой должно быть сообщение,
- ✓ непосредственно реквизит формы, к которому должно быть привязано сообщение.

Можно сделать все проще, если использовать метод УстановитьДанные объекта *СообщениеПользователю*. Тогда код с листинга 12.5 несколько упростится.

```
Если Не ЗначениеЗаполнено (Склад) Тогда
    Сообщение = Новый СообщениеПользователю;
    Сообщение.Текст = "Не заполнен склад";
    Сообщение.Поле = "Склад";
    Сообщение.УстановитьДанные (ЭтотОбъект);
    Сообщение.Сообщить ();
    Отказ = Истина;
КонецЕсли;
```

Листинг 12.6. Упрощенный вывод сообщения при проведении документа

Метод *УстановитьДанные* автоматически устанавливает свойства *ПутьКДанным* и *КлючДанных* на основании переданного в него объекта.

Пример 6.

При проведении документа *Продажа товара* реализовать следующую проверку: если в строке *Товары* количество равно 0, то отменить проведение, сообщить об этом пользователю и указать нужную строку.

Самостоятельно добавьте данный код в процедуру проведения документа

```
Для Каждого ТекСтрокаТовары Из Товары Цикл
    Если ТекСтрокаТовары.Количество = 0 Тогда
        ИндексСтроки = ТекСтрокаТовары.НомерСтроки - 1;
        Сообщение = Новый СообщениеПользователю;
        Сообщение.Текст = "Не заполнено количество";
        Сообщение.Поле = "Товары[" + ИндексСтроки + "].Количество";
        Сообщение.УстановитьДанные (ЭтотОбъект);
        Сообщение.Сообщить ();
        Отказ = Истина;
    КонецЕсли;
КонецЦикла;
```

Листинг 12.7. Проверка на количество при проведении документа